



Machine-Level Programming II: Control

15-213/18-213/14-513/15-513/18-613: Introduction to Computer Systems
6th Lecture, September 16, 2020

Announcements

■ Lab 1 (datalab)

- Due Thurs, Sept. 17, 11:59pm ET

■ Written Assignment 1 peer grading

- Due Wed, Sept. 23, 11:59pm ET

■ Written Assignment 2 available on [Canvas](#)

- Due Wed, Sept. 23, 11:59pm ET

■ Lab 2 (bomblab) will be available at midnight via [Autolab](#)

- Due Tues, Sept. 29, 11:59 pm ET

■ Recitation on bomblab this Monday

- In person: you have been contacted with your recitation info
- Online: use the zoom links provided on the Canvas homepage

Catching Up

- Reviewing LEAQ (based on after-class questions)
- Reviewing Arithmetic Expressions in ASM
- C -> Assembly -> Machine Code

LEA: Evaluate Memory Address Expression Without Accessing Memory

■ `leaq Src, Dst`

- `Src` is address computation expression \longrightarrow $D(Rb, Ri, S): \text{Reg}[Rb] + S * \text{Reg}[Ri] + D$
- Set `Dst` to address denoted by expression

■ Uses

- Computing address/pointer **WITHOUT ACCESSING MEMORY**
 - E.g., translation of `p = &x[i];`
- Compute arbitrary expressions of form: $b + (s * i) + d$, where $s = 1, 2, 4, \text{ or } 8$
 - [also w/o accessing memory]

■ Example

```
long m12(long x)
{
    return x*12;
}
```

Converted to ASM by compiler:

```
leaq (%rdi,%rdi,2), %rax    # t = x+2*x
salq $2, %rax               # return t<<2
```

LEA vs. other instructions (e.g., MOV)

■ `leaq D(Rb,Ri,S), dst`

- $dst \longleftarrow \text{Reg}[Rb] + S * \text{Reg}[Ri] + D$
- *NO MEMORY ACCESS HAPPENS!*

■ `movq D(Rb,Ri,S), dst`

- $dst \longleftarrow \text{Mem}[\text{Reg}[Rb] + S * \text{Reg}[Ri] + D]$
- *MEMORY ACCESS HAPPENS!*

Arithmetic Expression Example

```
long arith
(long x, long y, long z)
{
    long t1 = x+y;
    long t2 = z+t1;
    long t3 = x+4;
    long t4 = y * 48;
    long t5 = t3 + t4;
    long rval = t2 * t5;
    return rval;
}
```

arith:

```
leaq    (%rdi,%rsi), %rax
addq    %rdx, %rax
leaq    (%rsi,%rsi,2), %rdx
salq    $4, %rdx
leaq    4(%rdi,%rdx), %rcx
imulq   %rcx, %rax
ret
```

Interesting Instructions

- **leaq**: address computation
- **salq**: shift
- **imulq**: multiplication
 - Curious: only used once...

Understanding Arithmetic Expression

Example

```

long arith
(long x, long y, long z)
{
    long t1 = x+y;
    long t2 = z+t1;
    long t3 = x+4;
    long t4 = y * 48;
    long t5 = t3 + t4;
    long rval = t2 * t5;
    return rval;
}

```

arith:

```

leaq    (%rdi,%rsi), %rax    # t1
addq    %rdx, %rax          # t2
leaq    (%rsi,%rsi,2), %rdx
salq    $4, %rdx            # t4
leaq    4(%rdi,%rdx), %rcx   # t5
imulq   %rcx, %rax          # rval
ret

```

Register	Use(s)
%rdi	Argument x
%rsi	Argument y
%rdx	Argument z , t4
%rax	t1, t2, rval
%rcx	t5


$D(Rb, Ri, S): \text{Mem}[\text{Reg}[Rb] + S * \text{Reg}[Ri] + D]$

Today: Machine Programming I: Basics

- History of Intel processors and architectures
- Assembly Basics: Registers, operands, move
- Arithmetic & logical operations
- **C, assembly, machine code**

Turning C into Object Code

- Code in files `p1.c` `p2.c`
- Compile with command: `gcc -Og p1.c p2.c -o p`
 - Use basic optimizations (`-Og`) [New to recent versions of GCC]
 - Put resulting binary in file `p`



Compiling Into Assembly

C Code (sum.c)

```
long plus(long x, long y);

void sumstore(long x, long y,
              long *dest)
{
    long t = plus(x, y);
    *dest = t;
}
```

Generated x86-64 Assembly

```
sumstore:
    pushq    %rbx
    movq     %rdx, %rbx
    call     plus
    movq     %rax, (%rbx)
    popq     %rbx
    ret
```

Obtain (on shark machine) with command

```
gcc -Og -S sum.c
```

Produces file `sum.s`

Warning: Will get very different results on non-Shark machines (Andrew Linux, Mac OS-X, ...) due to different versions of gcc and different compiler settings.

What it really looks like

```
.globl  sumstore
.type   sumstore, @function
sumstore:
.LFB35:
    .cfi_startproc
    pushq   %rbx
    .cfi_def_cfa_offset 16
    .cfi_offset 3, -16
    movq    %rdx, %rbx
    call    plus
    movq    %rax, (%rbx)
    popq    %rbx
    .cfi_def_cfa_offset 8
    ret
    .cfi_endproc
.LFE35:
    .size   sumstore, .-sumstore
```

What it really looks like

Things that look weird
and are preceded by a '**'**
are generally directives.

```
.globl  sumstore
.type   sumstore, @function

sumstore:
.LFB35:
    .cfi_startproc
pushq   %rbx
.cfi_def_cfa_offset 16
.cfi_offset 3, -16
movq    %rdx, %rbx
call    plus
movq    %rax, (%rbx)
popq    %rbx
.cfi_def_cfa_offset 8
ret
.cfi_endproc

.LFE35:
.size   sumstore, .-sumstore
```

```
sumstore:
    pushq   %rbx
    movq    %rdx, %rbx
    call    plus
    movq    %rax, (%rbx)
    popq    %rbx
    ret
```

Object Code

Code for `sumstore`

0x0400595:

0x53

0x48

0x89

0xd3

0xe8

0xf2

0xff

0xff

0xff

0x48

0x89

0x03

0x5b

0xc3

- Total of 14 bytes
- Each instruction 1, 3, or 5 bytes
- Starts at address 0x0400595

■ Assembler

- Translates `.s` into `.o`
- Binary encoding of each instruction
- Nearly-complete image of executable code
- Missing linkages between code in different files

■ Linker

- Resolves references between files
- Combines with static run-time libraries
 - E.g., code for `malloc`, `printf`
- Some libraries are *dynamically linked*
 - Linking occurs when program begins execution

Machine Instruction Example

```
*dest = t;
```

```
movq %rax, (%rbx)
```

```
0x40059e:  48 89 03
```

■ C Code

- Store value `t` where designated by `dest`

■ Assembly

- Move 8-byte value to memory
 - Quad words in x86-64 parlance
- Operands:
 - `t`: Register `%rax`
 - `dest`: Register `%rbx`
 - `*dest`: Memory `M[%rbx]`

■ Object Code

- 3-byte instruction
- Stored at address `0x40059e`

Disassembling Object Code

Disassembled

```

0000000000400595 <sumstore>:
  400595:  53                      push    %rbx
  400596:  48 89 d3                mov     %rdx,%rbx
  400599:  e8 f2 ff ff ff         callq   400590 <plus>
  40059e:  48 89 03                mov     %rax, (%rbx)
  4005a1:  5b                      pop     %rbx
  4005a2:  c3                      retq

```

■ Disassembler

`objdump -d sum`

- Useful tool for examining object code
- Analyzes bit pattern of series of instructions
- Produces approximate rendition of assembly code
- Can be run on either a `.out` (complete executable) or `.o` file

Alternate Disassembly

Disassembled

```
Dump of assembler code for function sumstore:
0x0000000000400595 <+0>: push    %rbx
0x0000000000400596 <+1>: mov     %rdx,%rbx
0x0000000000400599 <+4>: callq   0x400590 <plus>
0x000000000040059e <+9>: mov     %rax, (%rbx)
0x00000000004005a1 <+12>: pop     %rbx
0x00000000004005a2 <+13>: retq
```

- **Within gdb Debugger**
 - Disassemble procedure
- ```
gdb sum
disassemble sumstore
```

# Alternate Disassembly

## Object Code

0x0400595:

0x53

0x48

0x89

0xd3

0xe8

0xf2

0xff

0xff

0xff

0x48

0x89

0x03

0x5b

0xc3

## Disassembled

Dump of assembler code for function sumstore:

```
0x0000000000400595 <+0>: push %rbx
0x0000000000400596 <+1>: mov %rdx,%rbx
0x0000000000400599 <+4>: callq 0x400590 <plus>
0x000000000040059e <+9>: mov %rax, (%rbx)
0x00000000004005a1 <+12>: pop %rbx
0x00000000004005a2 <+13>: retq
```

### ■ Within gdb Debugger

- Disassemble procedure

`gdb sum`

`disassemble sumstore`

- Examine the 14 bytes starting at `sumstore`

`x/14xb sumstore`

# What Can be Disassembled?

```
% objdump -d WINWORD.EXE
```

```
WINWORD.EXE: file format pei-i386
```

```
No symbols in "WINWORD.EXE".
```

```
Disassembly of section .text:
```

```
30001000 <.text>:
```

```
30001000:
```

```
30001001:
```

```
30001003:
```

```
30001005:
```

```
3000100a:
```

**Reverse engineering forbidden by  
Microsoft End User License Agreement**

- Anything that can be interpreted as executable code
- Disassembler examines bytes and reconstructs assembly source

# Machine Programming I: Summary

## ■ History of Intel processors and architectures

- Evolutionary design leads to many quirks and artifacts

## ■ C, assembly, machine code

- New forms of visible state: program counter, registers, ...
- Compiler must transform statements, expressions, procedures into low-level instruction sequences

## ■ Assembly Basics: Registers, operands, move

- The x86-64 move instructions cover wide range of data movement forms

## ■ Arithmetic

- C compiler will figure out different instruction combinations to carry out computation

# Today

- **Control: Condition codes**
- **Conditional branches**
- **Loops**
- **Switch Statements**

# Processor State (x86-64, Partial)

## ■ Information about currently executing program

- Temporary data  
( `%rax`, ... )
- Location of runtime stack  
( `%rsp` )
- Location of current code control point  
( `%rip`, ... )
- Status of recent tests  
( `CF`, `ZF`, `SF`, `OF` )

Current stack top

### Registers

|                   |                   |
|-------------------|-------------------|
| <code>%rax</code> | <code>%r8</code>  |
| <code>%rbx</code> | <code>%r9</code>  |
| <code>%rcx</code> | <code>%r10</code> |
| <code>%rdx</code> | <code>%r11</code> |
| <code>%rsi</code> | <code>%r12</code> |
| <code>%rdi</code> | <code>%r13</code> |
| <code>%rsp</code> | <code>%r14</code> |
| <code>%rbp</code> | <code>%r15</code> |

`%rip`

Instruction pointer

`CF`

`ZF`

`SF`

`OF`

Condition codes

# Condition Codes (Implicit Setting)

## ■ Single bit registers

- **CF**      Carry Flag (for unsigned)      **SF** Sign Flag (for signed)
- **ZF**      Zero Flag      **OF** Overflow Flag (for signed)

## ■ Implicitly set (as side effect) of arithmetic operations

Example: `addq Src, Dest`  $\leftrightarrow$  `t = a+b`

**CF set** if carry/borrow out from most significant bit (unsigned overflow)

**ZF set** if `t == 0`

**SF set** if `t < 0` (as signed)

**OF set** if two's-complement (signed) overflow

`(a>0 && b>0 && t<0) || (a<0 && b<0 && t>=0)`

## ■ Not set by `leaq` instruction

# ZF set when

000000000000...000000000000



# SF set when

$$\begin{array}{r} \text{yxxxxxxxxxxxxxxxxx} \dots \\ + \text{yxxxxxxxxxxxxxxxxx} \dots \\ \hline \text{1xxxxxxxxxxxxxxxxx} \dots \end{array}$$


For signed arithmetic, this reports when result is a negative number

# CF set when



For unsigned arithmetic, this reports overflow

# OF set when



$$z = \sim y$$

$(a > 0 \ \&\& \ b > 0 \ \&\& \ t < 0) \ || \ (a < 0 \ \&\& \ b < 0 \ \&\& \ t \geq 0)$

For signed arithmetic, this reports overflow

# Condition Codes (Explicit Setting: Compare)

## ■ Explicit Setting by Compare Instruction

- `cmpq Src2, Src1`
- `cmpq b, a` like computing `a-b` without setting destination
- **CF set** if carry/borrow out from most significant bit  
(used for unsigned comparisons)
- **ZF set** if `a == b`
- **SF set** if `(a-b) < 0` (as signed)
- **OF set** if two's-complement (signed) overflow  
`(a>0 && b<0 && (a-b)<0) || (a<0 && b>0 && (a-b)>0)`

# Condition Codes (Explicit Setting: Test)

## ■ Explicit Setting by Test instruction

- `testq Src2, Src1`
  - `testq b, a` like computing `a&b` without setting destination
- Sets condition codes based on value of `Src1` & `Src2`
- Useful to have one of the operands be a mask
- **ZF set** when `a&b == 0`
- **SF set** when `a&b < 0`

Very often:

```
testq %rax, %rax
```

# Condition Codes (Explicit Reading: Set)

## ■ Explicit Reading by Set Instructions

- **setX** *Dest*: Set low-order byte of destination *Dest* to 0 or 1 based on combinations of condition codes
- Does not alter remaining 7 bytes of *Dest*


| SetX         | Condition                  | Description               |
|--------------|----------------------------|---------------------------|
| <b>sete</b>  | <b>ZF</b>                  | Equal / Zero              |
| <b>setne</b> | <b>~ZF</b>                 | Not Equal / Not Zero      |
| <b>sets</b>  | <b>SF</b>                  | Negative                  |
| <b>setns</b> | <b>~SF</b>                 | Nonnegative               |
| <b>setg</b>  | <b>~ (SF^OF) &amp; ~ZF</b> | Greater (signed)          |
| <b>setge</b> | <b>~ (SF^OF)</b>           | Greater or Equal (signed) |
| <b>setl</b>  | <b>SF^OF</b>               | Less (signed)             |
| <b>setle</b> | <b>(SF^OF)   ZF</b>        | Less or Equal (signed)    |
| <b>seta</b>  | <b>~CF &amp; ~ZF</b>       | Above (unsigned)          |
| <b>setb</b>  | <b>CF</b>                  | Below (unsigned)          |

# Example: setl (Signed <)

## ■ Condition: $SF \wedge OF$

| SF | OF | $SF \wedge OF$ | Implication                                           |
|----|----|----------------|-------------------------------------------------------|
| 0  | 0  | 0              | No overflow, so SF implies not <                      |
| 1  | 0  | 1              | No overflow, so SF implies <                          |
| 0  | 1  | 1              | Overflow, so SF implies negative overflow, i.e. <     |
| 1  | 1  | 0              | Overflow, so SF implies positive overflow, i.e. not < |

negative overflow case



# x86-64 Integer Registers

|             |             |
|-------------|-------------|
| <b>%rax</b> | <b>%al</b>  |
| <b>%rbx</b> | <b>%bl</b>  |
| <b>%rcx</b> | <b>%cl</b>  |
| <b>%rdx</b> | <b>%dl</b>  |
| <b>%rsi</b> | <b>%sil</b> |
| <b>%rdi</b> | <b>%di</b>  |
| <b>%rsp</b> | <b>%spl</b> |
| <b>%rbp</b> | <b>%bpl</b> |

|             |              |
|-------------|--------------|
| <b>%r8</b>  | <b>%r8b</b>  |
| <b>%r9</b>  | <b>%r9b</b>  |
| <b>%r10</b> | <b>%r10b</b> |
| <b>%r11</b> | <b>%r11b</b> |
| <b>%r12</b> | <b>%r12b</b> |
| <b>%r13</b> | <b>%r13b</b> |
| <b>%r14</b> | <b>%r14b</b> |
| <b>%r15</b> | <b>%r15b</b> |

- Can reference low-order byte



# Explicit Reading Condition Codes (Cont.)


## ■ SetX Instructions:

- Set single byte based on combination of condition codes

## ■ One of addressable byte registers

- Does not alter remaining bytes
- Typically use **movzbl** to finish job
  - 32-bit instructions also set upper 32 bits to 0

```
int gt (long x, long y)
{
 return x > y;
}
```




```
cmpq %rsi, %rdi # Compare x:y
setg %al # Set when >
movzbl %al, %eax # Zero rest of %rax
ret
```

| Register | Use(s)            |
|----------|-------------------|
| %rdi     | Argument <b>x</b> |
| %rsi     | Argument <b>y</b> |
| %rax     | Return value      |

# Explicit Reading Condition Codes (Cont.)

Beware weirdness `movzbl` (and others)

`movzbl %al, %eax`



Zapped to all 0's

Use(s)

Argument **x**

Argument **y**

Return value

```
cmpq %rsi, %rdi # Compare x:y
setg %al # Set when >
movzbl %al, %eax # Zero rest of %rax
ret
```

# Today

- Control: Condition codes
- **Conditional branches**
- Loops
- Switch Statements

# Jumping

## ■ jX Instructions

- Jump to different part of code depending on condition codes
- Implicit reading of condition codes

| jX         | Condition                            | Description               |
|------------|--------------------------------------|---------------------------|
| <b>jmp</b> | 1                                    | Unconditional             |
| <b>je</b>  | ZF                                   | Equal / Zero              |
| <b>jne</b> | $\sim ZF$                            | Not Equal / Not Zero      |
| <b>js</b>  | SF                                   | Negative                  |
| <b>jns</b> | $\sim SF$                            | Nonnegative               |
| <b>jg</b>  | $\sim (SF \wedge OF) \ \& \ \sim ZF$ | Greater (signed)          |
| <b>jge</b> | $\sim (SF \wedge OF)$                | Greater or Equal (signed) |
| <b>jl</b>  | $SF \wedge OF$                       | Less (signed)             |
| <b>jle</b> | $(SF \wedge OF) \   \ ZF$            | Less or Equal (signed)    |
| <b>ja</b>  | $\sim CF \ \& \ \sim ZF$             | Above (unsigned)          |
| <b>jb</b>  | CF                                   | Below (unsigned)          |

# Conditional Branch Example (Old Style)

## ■ Generation

shark> gcc -Og -S **-fno-if-conversion** control.c

Get to this shortly

```
long absdiff
(long x, long y)
{
 long result;
 if (x > y)
 result = x-y;
 else
 result = y-x;
 return result;
}
```

absdiff:

```
 cmpq %rsi, %rdi # x:y, x-y
 jle .L4
 movq %rdi, %rax
 subq %rsi, %rax
 ret
.L4: # x <= y
 movq %rsi, %rax
 subq %rdi, %rax
 ret
```

| Register | Use(s)            |
|----------|-------------------|
| %rdi     | Argument <b>x</b> |
| %rsi     | Argument <b>y</b> |
| %rax     | Return value      |

# Expressing with Goto Code

- C allows goto statement
- Jump to position designated by label

```
long absdiff
(long x, long y)
{
 long result;
 if (x > y)
 result = x-y;
 else
 result = y-x;
 return result;
}
```

```
long absdiff_j
(long x, long y)
{
 long result;
 int ntest = (x <= y);
 if (ntest) goto Else;
 result = x-y;
 goto Done;
Else:
 result = y-x;
Done:
 return result;
}
```

# General Conditional Expression Translation (Using Branches)

## C Code

```
val = Test ? Then_Expr : Else_Expr;
```

```
val = x > y ? x - y : y - x;
```

## Goto Version

```
n timer = !Test;
if (ntest) goto Else;
val = Then_Expr;
goto Done;
Else:
 val = Else_Expr;
Done:
 . . .
```

- Create separate code regions for then & else expressions
- Execute appropriate one

# Using Conditional Moves

## ■ Conditional Move Instructions

- Instruction supports:  
if (Test) Dest  $\leftarrow$  Src
- Supported in post-1995 x86 processors
- GCC tries to use them
  - But, only when known to be safe

## ■ Why?

- Branches are very disruptive to instruction flow through pipelines
- Conditional moves do not require control transfer

## C Code

```
val = Test
 ? Then_Expr
 : Else_Expr;
```

## Goto Version

```
result = Then_Expr;
eval = Else_Expr;
nt = !Test;
if (nt) result = eval;
return result;
```



# Conditional Move Example

```

long absdiff
(long x, long y)
{
 long result;
 if (x > y)
 result = x-y;
 else
 result = y-x;
 return result;
}

```

| Register | Use(s)            |
|----------|-------------------|
| %rdi     | Argument <b>x</b> |
| %rsi     | Argument <b>y</b> |
| %rax     | Return value      |

absdiff:

```

movq %rdi, %rax # x
subq %rsi, %rax # result = x-y
movq %rsi, %rdx
subq %rdi, %rdx # eval = y-x
cmpq %rsi, %rdi # x:y
cmovle %rdx, %rax # if <=, result = eval
ret

```

When is  
this bad?

# Bad Cases for Conditional Move

## Expensive Computations

```
val = Test(x) ? Hard1(x) : Hard2(x);
```

Bad Performance

- Both values get computed
- Only makes sense when computations are very simple

## Risky Computations

```
val = p ? *p : 0;
```

Unsafe

- Both values get computed
- May have undesirable effects

## Computations with side effects

```
val = x > 0 ? x*=7 : x+=3;
```

Illegal

- Both values get computed
- Must be side-effect free

# Exercise

`cmpq b, a` like computing  $a - b$  w/o setting dest

- **CF set** if carry/borrow out from most significant bit (used for unsigned comparisons)
- **ZF set** if  $a == b$
- **SF set** if  $(a - b) < 0$  (as signed)
- **OF set** if two's-complement (signed) overflow

| SetX               | Condition                            | Description               |
|--------------------|--------------------------------------|---------------------------|
| <code>sete</code>  | ZF                                   | Equal / Zero              |
| <code>setne</code> | $\sim ZF$                            | Not Equal / Not Zero      |
| <code>sets</code>  | SF                                   | Negative                  |
| <code>setns</code> | $\sim SF$                            | Nonnegative               |
| <code>setg</code>  | $\sim (SF \wedge OF) \ \& \ \sim ZF$ | Greater (signed)          |
| <code>setge</code> | $\sim (SF \wedge OF)$                | Greater or Equal (signed) |
| <code>setl</code>  | $SF \wedge OF$                       | Less (signed)             |
| <code>setle</code> | $(SF \wedge OF) \mid ZF$             | Less or Equal (signed)    |
| <code>seta</code>  | $\sim CF \ \& \ \sim ZF$             | Above (unsigned)          |
| <code>setb</code>  | CF                                   | Below (unsigned)          |

```

xorq %rax, %rax
subq $1, %rax
cmpq $2, %rax
setl %al
movzblq %al, %eax

```

| %rax | SF | CF | OF | ZF |
|------|----|----|----|----|
|      |    |    |    |    |
|      |    |    |    |    |
|      |    |    |    |    |
|      |    |    |    |    |
|      |    |    |    |    |

Note: `setl` and `movzblq` do not modify condition codes

# Exercise

`cmpq b, a` like computing  $a - b$  w/o setting dest

- **CF set** if carry/borrow out from most significant bit (used for unsigned comparisons)
- **ZF set** if  $a == b$
- **SF set** if  $(a - b) < 0$  (as signed)
- **OF set** if two's-complement (signed) overflow

| SetX               | Condition                            | Description               |
|--------------------|--------------------------------------|---------------------------|
| <code>sete</code>  | ZF                                   | Equal / Zero              |
| <code>setne</code> | $\sim ZF$                            | Not Equal / Not Zero      |
| <code>sets</code>  | SF                                   | Negative                  |
| <code>setns</code> | $\sim SF$                            | Nonnegative               |
| <code>setg</code>  | $\sim (SF \wedge OF) \ \& \ \sim ZF$ | Greater (signed)          |
| <code>setge</code> | $\sim (SF \wedge OF)$                | Greater or Equal (signed) |
| <code>setl</code>  | $SF \wedge OF$                       | Less (signed)             |
| <code>setle</code> | $(SF \wedge OF) \mid ZF$             | Less or Equal (signed)    |
| <code>seta</code>  | $\sim CF \ \& \ \sim ZF$             | Above (unsigned)          |
| <code>setb</code>  | CF                                   | Below (unsigned)          |

```

xorq %rax, %rax
subq $1, %rax
cmpq $2, %rax
setl %al
movzblq %al, %eax

```

| %rax                  | SF | CF | OF | ZF |
|-----------------------|----|----|----|----|
| 0x0000 0000 0000 0000 | 0  | 0  | 0  | 1  |
| 0xFFFF FFFF FFFF FFFF | 1  | 1  | 0  | 0  |
| 0xFFFF FFFF FFFF FFFF | 1  | 0  | 0  | 0  |
| 0xFFFF FFFF FFFF FF01 | 1  | 0  | 0  | 0  |
| 0x0000 0000 0000 0001 | 1  | 0  | 0  | 0  |

Note: `setl` and `movzblq` do not modify condition codes

# Today

- Control: Condition codes
- Conditional branches
- **Loops**
- Switch Statements

# “Do-While” Loop Example

## C Code

```
long pcount_do
(unsigned long x) {
 long result = 0;
 do {
 result += x & 0x1;
 x >>= 1;
 } while (x);
 return result;
}
```

## Goto Version

```
long pcount_goto
(unsigned long x) {
 long result = 0;
loop:
 result += x & 0x1;
 x >>= 1;
 if(x) goto loop;
 return result;
}
```

- Count number of 1's in argument *x* (“popcount”)
- Use conditional branch to either continue looping or to exit loop

# “Do-While” Loop Compilation

```

long pcount_goto
(unsigned long x) {
 long result = 0;
loop:
 result += x & 0x1;
 x >>= 1;
 if(x) goto loop;
 return result;
}

```

| Register | Use(s)            |
|----------|-------------------|
| %rdi     | Argument <b>x</b> |
| %rax     | <b>result</b>     |

```

 movl $0, %eax # result = 0
.L2: # loop:
 movq %rdi, %rdx
 andl $1, %edx # t = x & 0x1
 addq %rdx, %rax # result += t
 shrq %rdi # x >>= 1
 jne .L2 # if(x) goto loop
 rep; ret

```

# Quiz Time!

Check out:

<https://canvas.cmu.edu/courses/17808>



# General “Do-While” Translation

## C Code

```
do
 Body
while (Test) ;
```

## Goto Version

```
loop:
 Body
 if (Test)
 goto loop
```

■ **Body:** {  
    Statement<sub>1</sub>;  
    Statement<sub>2</sub>;  
    ...  
    Statement<sub>n</sub>;  
}

# General “While” Translation #1

- “Jump-to-middle” translation
- Used with -Og

## While version

```
while (Test)
 Body
```



## Goto Version

```
 goto test;
loop:
 Body
test:
 if (Test)
 goto loop;
done:
```

# While Loop Example #1

## C Code

```
long pcount_while
(unsigned long x) {
 long result = 0;
 while (x) {
 result += x & 0x1;
 x >>= 1;
 }
 return result;
}
```

## Jump to Middle


```
long pcount_goto_jtm
(unsigned long x) {
 long result = 0;
 goto test;
loop:
 result += x & 0x1;
 x >>= 1;
test:
 if(x) goto loop;
 return result;
}
```

- Compare to do-while version of function
- Initial goto starts loop at test

# General “While” Translation #2

## While version

```
while (Test)
 Body
```



## Do-While Version

```
if (!Test)
 goto done;
do
 Body
 while (Test) ;
done:
```



## Goto Version

```
if (!Test)
 goto done;
loop:
 Body
 if (Test)
 goto loop;
done:
```

- “Do-while” conversion
- Used with -O1

# While Loop Example #2

## C Code

```
long pcount_while
(unsigned long x) {
 long result = 0;
 while (x) {
 result += x & 0x1;
 x >>= 1;
 }
 return result;
}
```

## Do-While Version

```
long pcount_goto_dw
(unsigned long x) {
 long result = 0;
 if (!x) goto done;
loop:
 result += x & 0x1;
 x >>= 1;
 if(x) goto loop;
done:
 return result;
}
```

- Initial conditional guards entrance to loop
- Compare to do-while version of function
  - Removes jump to middle. When is this good or bad?

# “For” Loop Form

## General Form

```
for (Init; Test; Update)
 Body
```

```
#define WSIZE 8*sizeof(int)
long pcount_for
(unsigned long x)
{
 size_t i;
 long result = 0;
 for (i = 0; i < WSIZE; i++)
 {
 unsigned bit =
 (x >> i) & 0x1;
 result += bit;
 }
 return result;
}
```

### Init

```
i = 0
```

### Test

```
i < WSIZE
```

### Update

```
i++
```

### Body

```
{
 unsigned bit =
 (x >> i) & 0x1;
 result += bit;
}
```

# “For” Loop → While Loop

## For Version

```
for (Init; Test; Update)
 Body
```



## While Version

```
Init;
while (Test) {
 Body
 Update;
}
```

# For-While Conversion

## Init

```
i = 0
```

## Test

```
i < WSIZE
```

## Update

```
i++
```

## Body

```
{
 unsigned bit =
 (x >> i) & 0x1;
 result += bit;
}
```

```
long pcount_for_while
 (unsigned long x)
{
 size_t i;
 long result = 0;
 i = 0;
 while (i < WSIZE)
 {
 unsigned bit =
 (x >> i) & 0x1;
 result += bit;
 i++;
 }
 return result;
}
```



# “For” Loop Do-While Conversion

## C Code

## Goto Version

```
long pcount_for
(unsigned long x)
{
 size_t i;
 long result = 0;
 for (i = 0; i < WSIZE; i++)
 {
 unsigned bit =
 (x >> i) & 0x1;
 result += bit;
 }
 return result;
}
```

- Initial test can be optimized away – **why?**

```
long pcount_for_goto_dw
(unsigned long x) {
 size_t i;
 long result = 0;
 i = 0; Init
 if (!(i < WSIZE)) ! Test
 goto done;
loop:
{
 unsigned bit =
 (x >> i) & 0x1; Body
 result += bit;
}
i++; Update
if (i < WSIZE) Test
 goto loop;
done:
 return result;
}
```

# Today

- Control: Condition codes
- Conditional branches
- Loops
- **Switch Statements**

```
long my_switch
(long x, long y, long z)
{
 long w = 1;
 switch(x) {
 case 1:
 w = y*z;
 break;
 case 2:
 w = y/z;
 /* Fall Through */
 case 3:
 w += z;
 break;
 case 5:
 case 6:
 w -= z;
 break;
 default:
 w = 2;
 }
 return w;
}
```

# Switch Statement Example

- **Multiple case labels**
  - Here: 5 & 6
- **Fall through cases**
  - Here: 2
- **Missing cases**
  - Here: 4

# Jump Table Structure

## Switch Form

```
switch(x) {
 case val_0:
 Block 0
 case val_1:
 Block 1
 . . .
 case val_n-1:
 Block n-1
}
```

## Jump Table

|       |         |
|-------|---------|
| jtab: | Targ0   |
|       | Targ1   |
|       | Targ2   |
|       | •       |
|       | •       |
|       | Targn-1 |

## Jump Targets

Targ0:

Code Block  
0

Targ1:

Code Block  
1

Targ2:

Code Block  
2

•  
•  
•

Targn-1:

Code Block  
n-1

## Translation (Extended C)

```
goto *JTab[x];
```

# Switch Statement Example

```

long my_switch
(long x, long y, long z)
{
 long w = 1;
 switch(x) {
 case 1:
.L3: w = y*z;
 break;
 case 2:
.L5: w = y/z;
 /* Fall Through */
 case 3:
.L9: w += z;
 break;
 case 5:
 case 6:
.L7: w -= z;
 break;
 default:
.L8: w = 2;
 }
 return w;
}

```

```

my_switch:
 cmpq $6, %rdi # x:6
 ja .L8 # if x > 6 jump
 # to default
 jmp *.L4(, %rdi, 8)

```

```

.section .rodata
 .align 8
.L4:
 .quad .L8 # x = 0
 .quad .L3 # x = 1
 .quad .L5 # x = 2
 .quad .L9 # x = 3
 .quad .L8 # x = 4
 .quad .L7 # x = 5
 .quad .L7 # x = 6

```

# Assembly Setup Explanation

## ■ Table Structure

- Each target requires 8 bytes
- Base address at `.L4`

## ■ Jumping

- **Direct:** `jmp .L8`
- Jump target is denoted by label `.L8`
- **Indirect:** `jmp *.L4(,%rdi,8)`
- Start of jump table: `.L4`
- Must scale by factor of 8 (addresses are 8 bytes)
- Fetch target from effective Address `.L4 + x*8`
  - Only for  $0 \leq x \leq 6$

## Jump table

```
.section .rodata
 .align 8
.L4:
 .quad .L8 # x = 0
 .quad .L3 # x = 1
 .quad .L5 # x = 2
 .quad .L9 # x = 3
 .quad .L8 # x = 4
 .quad .L7 # x = 5
 .quad .L7 # x = 6
```

# Code Blocks (x == 1)

```
switch(x) {
case 1: // .L3
 w = y*z;
 break;

 . . .
}
```

```
.L3:
 movq %rsi, %rax # y
 imulq %rdx, %rax # y*z
 ret
```

| Register | Use(s)            |
|----------|-------------------|
| %rdi     | Argument <b>x</b> |
| %rsi     | Argument <b>y</b> |
| %rdx     | Argument <b>z</b> |
| %rax     | Return value      |

# Handling Fall-Through

```
long w = 1;
.
.
.
switch(x) {
.
.
.
case 2:
 w = y/z;
 /* Fall Through */
case 3:
 w += z;
 break;
.
.
.
}
```

case 2:  
    w = y/z;  
    goto merge;

case 3:  
    w = 1;  
merge:  
    w += z;



# Code Blocks (x == 2, x == 3)

```

long w = 1;
. . .
switch(x) {
. . .
case 2:
 w = y/z;
 /* Fall Through */
case 3:
 w += z;
 break;
. . .
}

```

```

.L5: # Case 2
 movq %rsi, %rax
 cqto # sign extend
 # rax to rdx:rax
 idivq %rcx # y/z
 jmp .L6 # goto merge
.L9: # Case 3
 movl $1, %eax # w = 1
.L6: # merge:
 addq %rcx, %rax # w += z
 ret

```

| Register | Use(s)            |
|----------|-------------------|
| %rdi     | Argument <b>x</b> |
| %rsi     | Argument <b>y</b> |
| %rcx     | <b>z</b>          |
| %rax     | Return value      |

# Code Blocks (x == 5, x == 6, default)

```
switch(x) {
 . . .
 case 5: // .L7
 case 6: // .L7
 w -= z;
 break;
 default: // .L8
 w = 2;
}
```

```
.L7: # Case 5,6
 movl $1, %eax # w = 1
 subq %rdx, %rax # w -= z
 ret
.L8: # Default:
 movl $2, %eax # 2
 ret
```

| Register | Use(s)            |
|----------|-------------------|
| %rdi     | Argument <b>x</b> |
| %rsi     | Argument <b>y</b> |
| %rdx     | Argument <b>z</b> |
| %rax     | Return value      |

# Summarizing

## ■ C Control

- if-then-else
- do-while
- while, for
- switch

## ■ Assembler Control

- Conditional jump
- Conditional move
- Indirect jump (via jump tables)
- Compiler generates code sequence to implement more complex control

## ■ Standard Techniques

- Loops converted to do-while or jump-to-middle form
- Large switch statements use jump tables
- Sparse switch statements may use decision trees (if-elseif-elseif-else)

# Summary

## ■ Today

- Control: Condition codes
- Conditional branches & conditional moves
- Loops
- Switch statements

## ■ Next Time

- Stack
- Call / return
- Procedure call discipline

# Finding Jump Table in Binary

```

00000000004005e0 <switch_eg>:
4005e0: 48 89 d1 mov %rdx,%rcx
4005e3: 48 83 ff 06 cmp $0x6,%rdi
4005e7: 77 2b ja 400614 <switch_eg+0x34>
4005e9: ff 24 fd f0 07 40 00 jmpq *0x4007f0(,%rdi,8)
4005f0: 48 89 f0 mov %rsi,%rax
4005f3: 48 0f af c2 imul %rdx,%rax
4005f7: c3 retq
4005f8: 48 89 f0 mov %rsi,%rax
4005fb: 48 99 cqto
4005fd: 48 f7 f9 idiv %rcx
400600: eb 05 jmp 400607 <switch_eg+0x27>
400602: b8 01 00 00 00 mov $0x1,%eax
400607: 48 01 c8 add %rcx,%rax
40060a: c3 retq
40060b: b8 01 00 00 00 mov $0x1,%eax
400610: 48 29 d0 sub %rdx,%rax
400613: c3 retq
400614: b8 02 00 00 00 mov $0x2,%eax
400619: c3 retq

```

# Finding Jump Table in Binary (cont.)

```
00000000004005e0 <switch_eg>:
. . .
4005e9: ff 24 fd f0 07 40 00 jmpq *0x4007f0(,%rdi,8)
. . .
```

```
% gdb switch
(gdb) x /8xg 0x4007f0
0x4007f0: 0x0000000000400614 0x00000000004005f0
0x400800: 0x00000000004005f8 0x0000000000400602
0x400810: 0x0000000000400614 0x000000000040060b
0x400820: 0x000000000040060b 0x2c646c25203d2078
(gdb)
```

# Finding Jump Table in Binary (cont.)

```
% gdb switch
(gdb) x /8xg 0x4007f0
0x4007f0: 0x000000000000400614 0x0000000000004005f0
0x400800: 0x0000000000004005f8 0x000000000000400602
0x400810: 0x000000000000400614 0x00000000000040060b
0x400820: 0x00000000000040060b 0x2c646c25203d2078
```

```
. . .
4005f0: 48 89 f0 mov %rsi,%rax
4005f3: 48 0f af c2 imul %rdx,%rax
4005f7: c3 retq
4005f8: 48 89 f0 mov %rsi,%rax
4005fb: 48 99 cqto
4005fd: 48 f7 f9 idiv %rcx
400600: eb 05 jmp 400607 <switch_eg+0x27>
400602: b8 01 00 00 00 mov $0x1,%eax
400607: 48 01 c8 add %rcx,%rax
40060a: c3 retq
40060b: b8 01 00 00 00 mov $0x1,%eax
400610: 48 29 d0 sub %rdx,%rax
400613: c3 retq
400614: b8 02 00 00 00 mov $0x2,%eax
400619: c3 retq
```